

FILLPATTERNS in SGPLOT Graphs

Pankhil Shah, PPD, Morrisville, NC

ABSTRACT

With more updates to PROC SGPLOT in SAS 9.3, there has been a substantial change in graph programming. Programming code size and complexity have been reduced compared to PROC GPLOT/GCHART, and with little effort, one can create much better quality graphs with PROC SGPLOT. However, this transition has a few down falls, as PROC SGPLOT doesn't share some of the useful features of PROC GPLOT/GCHART. One of them is showing fill-patterns in graphs produced by PROC GPLOT/GCHART. Currently, PROC SGPLOT doesn't support any direct styling option for showing patterns to differentiate between groups. Also, styling using SG Attribute Map data set used in SG procedures is not useful in the same regards. The main purpose of this paper is to provide a work-around on the above mentioned issue by creating new or modifying an existing SAS predefined templates for SG procedures.

INTRODUCTION

Fill-patterns help distinguish between different categories when producing or printing black and white outputs. The purpose of this paper is to demonstrate how one can show FILLPATTERNS in a bar chart using PROC SGPLOT. Currently, PROC SGPLOT doesn't support any direct styling option for showing patterns to differentiate between groups. Also, styling using SG Attribute Map data set used in SG procedures is not useful in the same regards. This can be achieved by a simple modification of a predefined template or creating a new template with styling features as will be discussed. This procedure is equivalent to the PATTERN statement in PROC GPLOT/GCHART.

Before going into more detail regarding a template update, let's review the PATTERN statement:

The PATTERN statement defines patterns and colors for the graph. It is of the form:

PATTERN*n* **COLOR**=*color* **VALUE**=*pattern* **REPEAT**=*m*;

Where *n* can be any number from 1 to 99. If *n* is not specified, 1 is assumed.

The **COLOR**= option specifies a valid color of the device.

The **VALUE**= option specifies the pattern to use for the bars. Some valid values for bar charts are:

EMPTY	requests an empty pattern (abbreviated as E)
SOLID	requests a solid pattern (abbreviated as S)
<i>Xn</i>	draws crosshatched lines of density <i>n</i> , <i>n</i> = 1,2,3,4,5
<i>Ln</i>	draws left-slanting lines of density <i>n</i> , <i>n</i> =1,2,3,4,5
<i>Rn</i>	draws right-slanting lines of density <i>n</i> , <i>n</i> =1,2,3,4,5

The **REPEAT**= option specifies the number of times a PATTERN is applied before using the next PATTERN statement.

The PROC TEMPLATE update in SAS 9.3 for SGPLOT will specify COLOR and VALUE for bar charts, similar to the PATTERN statement.

BAR CHART USING PROC GCHART

Let's take a look at a hypothetical clinical scenario where we want to plot means and their 95% confidence interval of Leukocytes counts by treatment group.

The following code will create the Vertical Bar chart shown in Figure 1.

```

AXIS1  order=0 to 20 by 2
       offset= (0,0)
       value=(h=9pt)
       minor=none
       label=(A=90 R=0 h=11pt j=c "Mean (95% CI) of Leukocytes (10E9/L)");

AXIS2  label=none
       value=none
       nobrackets
       origin=(3 cm,)
       interval=even
       offset=(2 cm,2 cm);

AXIS3  label=none
       value=(h=9pt)
       nobrackets;

LEGEND1 label=("Treatment:");

proc gchart data=<dataset>;
       format aval best.;
vbar trt01pn /      discrete      group = newwk      subgroup = char_trt
                   gspace = 3      space = 0          width = 3.5
                   coutline = black type = mean        sumvar = aval
                   raxis = axis1    maxis = axis2      gaxis = axis3
                   legend = legend1 patternid = subgroup;

       pattern1 value=e ;
       pattern2 value=r2 color=cxB2182B;
       pattern3 value=x2 color= cx01665E;

run;
quit;

```




Figure 1. VBAR chart created using PROC GCHART

BAR CHART USING PROC SGPLOT

Now we will duplicate the above figure using PROC SGPLOT.

Below is showing how to modify an existing template by adding GRAPHBAR style element (shown in bold red box below) for the fill pattern. Here we are creating MYTEMPLATE style from the existing DEFAULT style.

The following code will create the Vertical Bar chart shown in Figure 2.

```
proc template;
  define style mytemplate ;
    parent=styles.default;

    /* To display fill-patterns, the GraphBar style
       element must be included with the FILLPATTERN
       as an option. */

    style GraphBar from GraphComponent /
      displayopts = "outline fillpattern";

    /* Define fill-patterns by using the FILLPATTERN
       style element attribute. */

    style GraphData1 from GraphData1 /
      fillpattern = "E";
    style GraphData2 from GraphData2 /
      fillpattern = "R2";
    style GraphData3 from GraphData3 /
      fillpattern = "X2";
    style GraphData4 from GraphData4 /
      fillpattern = "L2";
    style GraphData5 from GraphData5 /
      fillpattern = "S";

    <more lines of code...>
run;
```

Now, incorporating the template in the ODS RTF destination:

```
ods rtf file = <filename> style = mytemplate;

proc sgplot data = <dataset> dattrmap = myattrmap;

  vbar newwk /          response = aval          group = char_trt
                      stat = mean             groupdisplay = cluster
                      grouporder = data      limitstat = stddev
                      attrid = trt           barwidth = 1.0
                      clusterwidth = 0.4    limits = both
                      name = "sub";
  xaxis               label = "Week"          values = (0 12 24);

  yaxis               label = "Mean (95% CI) of Leukocytes (10E9/L)"
                      values = (0 to 20 by 2) ;

  keylegend "sub" /   title = "Treatment:";
run;

ods rtf close;
```




Figure 2. Showing FILLPATTERNS by updating PROC TEMPLATE for PROC SGPLOT

As we can see, updating TEMPLATE will produce a similar result as shown in Figure 1. Additionally, other attributes like line thickness and contrast can be specified in the pattern to make the groups look easily distinguishable. Below examples demonstrates use of a CONTRASTCOLOR style element attribute in GraphData style to change line color (shown below in Figure 3).

```

style GraphData1 from GraphData1 /
  fillpattern = "E";
style GraphData2 from GraphData2 /
  fillpattern = "R2"
  contrastcolor = cx363636;
style GraphData3 from GraphData3 /
  fillpattern = "X2"
  contrastcolor = cxCCCCCC;
style GraphData4 from GraphData4 /
  fillpattern = "L2"
  contrastcolor = cxBDBDBD;
style GraphData5 from GraphData5 /
  fillpattern = "S"
  contrastcolor = cxgrayaa;
    
```





Figure 3. Changing line colors using CONTRASTCOLOR style element attribute in PROC TEMPLATE

NOTE:

- DO NOT USE **NOFILL** or **NOOUTLIE** option in SG procedure. Otherwise, it will overwrite the template and will not produce any pattern.
- FILLPATTERN style element attribute takes the same values as VALUE option in PATTERN statement. Below is the schematic of all of the patterns available for bars and blocks.



Display 1. Schematic of all of the patterns available for bars and blocks

- The fill patterns appear for grouped data in the following types of plots, whether generated using the SGPLOT or SG PANEL procedure:
 HBAR and VBAR
 HBARPARM and VBARPARM

- Some SAS predefined templates in the template catalog can be used as good examples for producing grouped bars with fill pattern as below:

JOURNAL2
JOURNAL3
MONOCHROMEPRINTER

CONCLUSION

In summary, updating STYLE in PROC TEMPLATE is the only way to show **fill-patterns** in graphs when using SG procedures. This approach is recommended over using PATTERN statement in PROC GCHART because of its ease to implement and improved graph resolution. In addition, the ability to create a centralized template and call it across logically-related outputs will enable overall consistency in results.

REFERENCES

<http://support.sas.com/documentation/cdl/en/grstatproc/65235/HTML/default/viewer.htm#p0relmtk2s8ac9n13bnmicsgsmwu.htm>

<http://stat.psu.edu/~hma/PSU/defense/graph/sasgraph.pdf>.

<http://support.sas.com/kb/45/663.html>

RECOMMENDED READING

- Base SAS[®] Procedures Guide
- SAS[®] For Dummies[®]
- SAS[®] SG Procedure

ACKNOWLEDGMENTS

I am grateful to Abhinav Srivastva, Ken Borowiak, Thomas Souers, Karen Easterling and Kunal Agnihotri for their careful review and valuable comments.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Pankhil Shah
Enterprise: PPD
Address: 3900 Paramount Pkwy
City, State ZIP: Morrisville, NC 27560
Work Phone: 919-456-4694
E-mail: pankhil.shah@ppdi.com

DISCLAIMER

The contents of this paper are the work of the author and do not necessarily represent the opinions, recommendations, or practices of PPD

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.